
Vorwort

DieseDokumentation beschreibt inNatural eingebaute Funktionen, die in bestimmten Statements
benutzt werden können, siehe Systemfunktionen im Leitfaden zur Programmierung.

Diese Dokumentation ist in die folgenden Abschnitte untergliedert:

Beschreibt Systemfunktionen, die im Zusammenhang mit einer
Programmschleife verwendet werden können.

Systemfunktionen für
Verarbeitungsschleifen

Beschreibt Systemfunktionen, die in arithmetischen Statements und in
logischen Bedingungen (Logical Condition Criteria, LCC) unterstützt
werden.

Mathematische
Systemfunktionen

Beschreibt Systemfunktionen zurAuswertungdesMinimums/Maximums
eines Feldes; Systemfunktion zur Feldidentifizierung; Systemfunktion zum

Verschiedene
Systemfunktionen

Empfangen des Return Codes von einem Nicht-Natural-Programm;
Systemfunktion zum Konvertieren von „nicht richtig sortierten“ Zeichen
in andere Zeichen, die vom Sortierprogramm oder Datenbanksystem
alphabetisch „richtig sortiert“ werden; Systemfunktion zur Umsetzung
von Klein- auf Großbuchstaben; Systemfunktion zum Entfernen von
führenden und/oder nachfolgenden Leerzeichen.

Beschreibt Funktionen,, die als Natural-Objekte ausgeliefert werden, um
zum Beispiel die URL-Kodierung und die Base64-Konvertierung zu
unterstützen.

Funktionen, die als
Natural-Objekte ausgeliefert
werden

Siehe auch Beispiel für Systemvariablen und Systemfunktionen im Leitfaden zur Programmierung:

v

Dokumentationskonventionen

BeschreibungKonvention

>Kennzeichnet Elemente auf einem Bildschirm.Fettschrift

Kennzeichnet Namen und Orte von Diensten im Format
Ordner.Unterordner.Dienst, Programmierschnittstellen (APIs), Namen von
Klassen, Methoden und Properties in Java.

Nichtproportionale
Schrift

Kennzeichnet:Kursivschrift

Variablen, für die Sie situations- oder umgebungsspezifische Werte angeben
müssen.
Neue Begriffe, wenn sie erstmals im Text auftreten.
Verweise auf andere Dokumentationsquellen.

Kennzeichnet:Nichtproportionale
Schrift

Text, den Sie eingeben müssen.
Meldungen, die vom System angezeigt werden.
Programmcode.

Zeigt eine Reihe von Auswahlmöglichkeiten an, von denen Sie eine auswählen
müssen. Geben Sie nur die innerhalb der geschweiften Klammern vorhandenen
Informationen ein. Geben Sie nicht die Klammersymbole { } ein.

{ }

Trennt zwei sich gegenseitig ausschließende Auswahlmöglichkeiten in einer
Syntaxzeile voneinander ab. Geben Sie eine derAuswahlmöglichkeiten ein. Geben
Sie nicht das Symbol | ein.

|

Zeigt eine oder mehrere Optionen an. Geben Sie nur die innerhalb der eckigen
Klammern vorhandenen Informationen ein. Geben Sie nicht die Klammersymbole
[ ] ein.

[ ]

Zeigt an, dass Sie mehrere Auswahlmöglichkeiten desselben Typs eingeben
können. Geben Sie nur die Informationen ein. Geben Sie nicht die drei
Auslassungspunkte (...) ein.

...

Online-Informationen und Support

Produktdokumentation

Sie finden die Produktdokumentation auf unsererDokumentationswebsite unterhttps://documen-
tation.softwareag.com.

Zusätzlich können Sie auch über https://www.softwareag.cloud auf die Dokumentation für die
Cloud-Produkte zugreifen. Navigieren Sie zumgewünschten Produkt und gehen Sie dann, je nach
Produkt, zu „Developer Center“, „User Center“ oder „Documentation“.

Produktschulungen

Sie finden hilfreiches Produktschulungsmaterial auf unserem Lernportal unter https://knowled-
ge.softwareag.com.

Tech Community

Auf der Website unserer Tech Community unter https://techcommunity.softwareag.com können
Sie mit Experten der Software AG zusammenarbeiten. Von hier aus können Sie zum Beispiel:

■ Unsere umfangreiche Wissensdatenbank durchsuchen.
■ In unseren Diskussionsforen Fragen stellen und Antworten finden.
■ Die neuesten Nachrichten und Ankündigungen der Software AG lesen.
■ Unsere Communities erkunden.
■ Unsere öffentlichen Repositories auf GitHub and Docker unter https://github.com/softwareag
und https://hub.docker.com/publishers/softwareag besuchen und weitere Ressourcen der
Software AG entdecken.

Produktsupport

Support für die Produkte der Software AG steht lizenzierten Kunden über unser Empower-Portal
unter https://empower.softwareag.com zur Verfügung. Für viele Dienstleistungen auf diesem
Portal benötigen Sie ein Konto. Wenn Sie noch keines haben, dann können Sie es unter htt-
ps://empower.softwareag.com/register beantragen. Sobald Sie ein Konto haben, können Sie zum
Beispiel:

■ Produkte, Aktualisierungen und Programmkorrekturen herunterladen.
■ Das Knowledge Center nach technischen Informationen und Tipps durchsuchen.
■ Frühwarnungen und kritische Alarme abonnieren.
■ Supportfälle öffnen und aktualisieren.
■ Anfragen für neue Produktmerkmale einreichen.

Datenschutz

Die Produkte der Software AG stellen Funktionen zur Verarbeitung von personenbezogenen
DatengemäßderDatenschutz-Grundverordnung (DSGVO)derEuropäischenUnionzurVerfügung.
Gegebenenfalls sind in der betreffenden Systemverwaltungsdokumentation entsprechende
Schritte dokumentiert.

DiesesKapitel erläutert dieNatural-Systemfunktionen, die imZusammenhangmit einer Programm-
schleife verwendet werden können.

Systemfunktionen für Verarbeitungsschleifen benutzen

■ Spezifikation/Auswertung
■ Systemfunktionen im SORT GIVE-Statement
■ Arithmetischer Überlauf bei AVER, NAVER, SUM oder TOTAL
■ Statement-Referenzierung (r)

Spezifikation/Auswertung

Natural-Systemfunktionen können angegeben werden in

■ zuweisenden und arithmetischen Statements:
■ MOVE

■ ASSIGN

■ COMPUTE

■ ADD

■ SUBTRACT

■ MULTIPLY

■ DIVIDE

■ in Eingabe-Ausgabe-Statements:
■ DISPLAY

■ PRINT

■ WRITE

die in einem der folgenden Statement-Blöcke stehen:

■ AT BREAK

■ AT END OF DATA

■ AT END OF PAGE

d.h. für alle Verarbeitungsschleifen in den Statements FIND, READ, HISTOGRAM, SORT oder READ WORK
FILE.

Wenn eine Systemfunktion in einem AT END OF PAGE-Statement verwendet wird, muss das
betreffende DISPLAY-Statement eine GIVE SYSTEM FUNCTIONS-Klausel enthalten.

Datensätze, die aufgrund einer WHERE-Klausel zurückgewiesenwerden,werden von einer System-
funktion nicht ausgewertet.

Wenn mittels Systemfunktionen Datenbankfelder ausgewertet werden, die aus FIND-, READ-,
HISTOGRAM- oder SORT-Schleifen stammen, die auf verschiedenen Ebenen liegen, werden die
Feldwerte jeweils entsprechend ihrer Position in der Verarbeitungsschleifen-Hierarchie verarbeitet.
Werte einer äußeren Schleife werden zum Beispiel nur dann verarbeitet, wenn für diese Schleife
neue Datenwerte erhalten werden.

Wird eine Benutzervariable vor der ersten Verarbeitungsschleife definiert, wird sie für System-
funktionen in der Schleife ausgewertet, in der das AT BREAK-, AT END OF DATA- oder AT END OF
PAGE-Statement steht; wird eine Benutzervariable innerhalb einer Schleife definiert, wird sie in
der gleichen Weise wie ein Datenbankfeld in der derselben Schleife behandelt.

Bei demselektivenEinsatz von Systemfunktionen zurAuswertungvonBenutzervariablen empfiehlt
es sich, eine bestimmte Verarbeitungsschleife zu referenzieren (mittels Statement-Label oder
Quellcode-Zeilennummer), um genau festzulegen, in welcher Schleife der Wert der Benutzerva-
riablen ausgewertet werden soll.

Systemfunktionen im SORT GIVE-Statement

Eine Systemfunktion kann auch referenziert werden, nachdem sie in der GIVE-Klausel eines
SORT-Statements ausgewertet wurde.

In diesem Fall muss dem Namen der Systemfunktion bei der Referenzierung ein Stern (*) voran-
gestellt werden.

Arithmetischer Überlauf bei AVER, NAVER, SUM oder TOTAL

Ein Feld, das Sie mit den Systemfunktionen AVER, NAVER, SUM oder TOTAL verwenden, muss lang
genug sein (standardmäßig oder vomBenutzer definiert), umdie Summeder Feldwerte aufzuneh-
men. Falls der addierteWert die Länge des Feldes überschreitet, erhalten Sie eine Fehlermeldung.

Normalerweise hat die Systemfunktion dieselbe Länge wie das angegebene Feld; ist diese Länge
nicht ausreichend, dann verwenden Sie den Parameter NL des SORT GIVE-Statements, um die
Ausgabelänge zu vergrößern:

SUM(field)(NL=nn)

Dadurch vergrößert sich nicht nur die Ausgabelänge, sondern auch die interne Länge des Feldes.

Statement-Referenzierung (r)

Statement-Referenzierung kann auch auf bei Systemfunktionen angewendet werden. Weitere
Einzelheiten entnehmen Sie dem Unterabschnitt Datenbankfelder mit der (r)-Notation referenzieren
- Notation (r) im Abschnitt Benutzervariablen im Natural Leitfaden zur Programmierung.

DurchVerwendung eines Statement-Labels oderAngabe derQuellcode-Zeilennummer (r) können
Sie bestimmen, in welcher Verarbeitungsschleife die jeweilige Systemfunktion für das betreffende
Feld ausgewertet werden soll.

AVER(r)(field)

Wie Feld.

Ausnahme: bei einem Feldmit Format N hat AVER(field) Format P (mit derselben Länge
wie das Feld).

Format/Länge:

Diese Systemfunktion enthält den Durchschnittswert (Average) aller Werte des angegebenen
Feldes. AVERwird aktualisiert, wenn die Bedingung, unter der AVER angefordert wurde, erfüllt ist.

COUNT(r)(field)

P7Format/Länge:

Diese Systemfunktion zählt dieDurchläufe einer Verarbeitungsschleife. DerWert von COUNT erhöht
sich jedesmal um 1, wenn die Verarbeitungsschleife, in der sich COUNT befindet, durchlaufenwird,
und zwar unabhängig vomWert des mit COUNT angegebenen Feldes.

MAX(r)(field)

Wie Feld.Format/Länge:

Diese Systemfunktion enthält den größtenWert des angegebenen Feldes. MAXwird (falls erforder-
lich) jedesmal aktualisiert, wenn die Verarbeitungsschleife, in der sich MAX befindet, ausgeführt
wird.

MIN(r)(field)

Wie Feld.Format/Länge:

Diese Systemfunktion enthält den kleinsten Wert des angegebenen Feldes. MINwird (falls erfor-
derlich) jedesmal aktualisiert, wenn die Verarbeitungsschleife, in der sich MIN befindet, ausgeführt
wird.

NAVER(r)(field)

Wie Feld.

Ausnahme: bei einemFeldmit FormatNhat NAVER(field) Format P (mit derselben Länge
wie das Feld).

Format/Länge:

Diese Systemfunktion enthält den Durchschnittswert (Average) aller Werte des angegebenen
Feldes,wobeiNullwerte nicht berücksichtigtwerden. NAVERwird aktualisiert, wenndie Bedingung,
unter der NAVER angefordert wurde, erfüllt ist.

NCOUNT(r)(field)

P7Format/Länge:

Diese Systemfunktion zählt die Durchläufe einer Verarbeitungsschleife. Der Wert von NCOUNT
erhöht sich jedesmal um1,wenndieVerarbeitungsschleife, in der sich NCOUNT befindet, durchlaufen
wird, wobei Durchläufe, bei denen der Wert des angegebenen Feldes Null ist, nicht mitgezählt
werden.

Ob das Ergebnis von NCOUNT ein Array oder ein Skalarwert ist, hängt vom Argument (field) ab.
Die Anzahl der resultierenden Ausprägungen ist dieselbe wie bei Feld.

11Systemfunktionen

NMIN(r)(field)

Wie Feld.Format/Länge:

Diese Systemfunktion enthält den kleinstenWert des angegebenen Feldes, wobei Nullwerte nicht
berücksichtigtwerden. NMINwird (falls erforderlich) jedesmal aktualisiert, wenn die Verarbeitungs-
schleife, in der sich NMIN befindet, ausgeführt wird.

OLD(r)(field)

Wie Feld.Format/Länge:

Diese Systemfunktion enthält den „alten“ Wert des angegebenen Feldes, d.h. den Wert, den das
Feld vor einem in einer AT BREAK-Bedingung spezifizierten Gruppenwechsel (Wechsel des Feld-
wertes) bzw. vor einer Seitenende- oder Datenende-Bedingung (END OF PAGE, END OF DATA) hatte.

SUM(r)(field)

Wie Feld.

Ausnahme: bei einem Feld mit Format N hat SUM(field) Format P (mit derselben Länge
wie das Feld).

Format/Länge:

Diese Systemfunktion enthält die Summe aller Werte des angegebenen Feldes. SUMwird jedesmal
aktualisiert, wenn die Verarbeitungsschleife, in der sich SUM befindet, ausgeführt wird. SUMwird
nach jedem AT BREAK-Gruppenwechsel wieder auf Null gesetzt, addiert also nur Werte zwischen
zwei Gruppenwechseln.

TOTAL(r)(field)

Wie Feld.

Ausnahme: bei einemFeldmit FormatNhat TOTAL(field) Format P (mit derselben Länge
wie das Feld).

Format/Länge:

Diese Systemfunktion enthält die Gesamtsumme aller Werte des angegebenen Feldes in allen
offenen Verarbeitungsschleifen, in denen TOTAL vorkommt.

Beispiele

■ Beispiel 1 − AT BREAK-Statement mit Natural-Systemfunktionen OLD, MIN, AVER, MAX, SUM,
COUNT
■ Beispiel 2 − AT BREAK-Statement mit Natural-Systemfunktion AVER
■ Beispiel 3 − AT END OF DATA-Statement mit Systemfunktionen MAX, MIN, AVER
■ Beispiel 4 − AT END OF PAGE-Statement mit Systemfunktion AVER

Beispiel 1 − AT BREAK-Statement mit Natural-Systemfunktionen OLD, MIN, AVER, MAX, SUM, COUNT

** Example 'ATBEX3': AT BREAK (with Natural system functions)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 SALARY (1)
2 CURR-CODE (1)

END-DEFINE
*
LIMIT 3
READ EMPLOY-VIEW LOGICAL BY CITY = 'SALT LAKE CITY'

DISPLAY NOTITLE CITY NAME 'SALARY' SALARY(1) 'CURRENCY' CURR-CODE(1)
/*
AT BREAK OF CITY

WRITE / OLD(CITY) (EM=X^X^X^X^X^X^X^X^X^X^X^X^X^X^X)
31T ' MINIMUM:' MIN(SALARY(1)) CURR-CODE(1) /
31T ' AVERAGE:' AVER(SALARY(1)) CURR-CODE(1) /
31T ' MAXIMUM:' MAX(SALARY(1)) CURR-CODE(1) /
31T ' SUM:' SUM(SALARY(1)) CURR-CODE(1) /
35T COUNT(SALARY(1)) 'RECORDS FOUND' /

END-BREAK                                           
  /*                                                  
  AT END OF DATA                                      

    WRITE 22T 'TOTAL (ALL RECORDS):'                  
              T*SALARY TOTAL(SALARY(1))  CURR-CODE(1) 
  END-ENDDATA                                         
END-READ                                              
*                                                     
END      ↩

Ausgabe des Programms ATBEX3:

CITY                 NAME           SALARY   CURRENCY
-------------------- -------------------- ---------- --------
                                                             
SALT LAKE CITY       ANDERSON                  50000 USD     
SALT LAKE CITY       SAMUELSON                 24000 USD     
                                                             
S A L T   L A K E   C I T Y      MINIMUM:      24000 USD     
                                 AVERAGE:      37000 USD     
                                 MAXIMUM:      50000 USD     
                                     SUM:      74000 USD     
                                         2 RECORDS FOUND     
                                                             
SAN DIEGO            GEE                       60000 USD     
                                                             
S A N   D I E G O                MINIMUM:      60000 USD     
                                 AVERAGE:      60000 USD     
                                 MAXIMUM:      60000 USD     
                                     SUM:      60000 USD     
                                         1 RECORDS FOUND     
                                                             
                     TOTAL (ALL RECORDS):     134000 USD    ↩

Beispiel 2 − AT BREAK-Statement mit Natural-Systemfunktion AVER

** Example 'ATBEX4': AT BREAK (with Natural system functions)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 SALARY (2)

*
1 #INC-SALARY (P11)
END-DEFINE
*
LIMIT 4
EMPL. READ EMPLOY-VIEW BY CITY STARTING FROM 'ALBU'

COMPUTE #INC-SALARY = SALARY (1) + SALARY (2)
DISPLAY NAME CITY SALARY (1:2) 'CUMULATIVE' #INC-SALARY
SKIP 1
/*
AT BREAK CITY

Systemfunktionen14

Systemfunktionen für Verarbeitungsschleifen



WRITE NOTITLE
'AVERAGE:' T*SALARY (1) AVER(SALARY(1)) /
'AVERAGE CUMULATIVE:' T*#INC-SALARY AVER(EMPL.) (#INC-SALARY)

END-BREAK                                                        
END-READ                                                           
*                                                                  
END                                                        ↩

Ausgabe des Programms ATBEX4:

NAME CITY ANNUAL CUMULATIVE
SALARY

-------------------- -------------------- ---------- ------------

HAMMOND ALBUQUERQUE 22000 42200
20200

ROLLING ALBUQUERQUE 34000 65200
31200

FREEMAN ALBUQUERQUE 34000 65200
31200

LINCOLN ALBUQUERQUE 41000 78700
37700

AVERAGE: 32750
AVERAGE CUMULATIVE: 62825

Beispiel 3 − AT END OF DATA-Statement mit Systemfunktionen MAX, MIN, AVER

** Example 'AEDEX1S': AT END OF DATA
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 FIRST-NAME
2 SALARY (1)
2 CURR-CODE (1)

END-DEFINE
*
LIMIT 5
EMP. FIND EMPLOY-VIEW WITH CITY = 'STUTTGART'

IF NO RECORDS FOUND
ENTER

END-NOREC
DISPLAY PERSONNEL-ID NAME FIRST-NAME

SALARY (1) CURR-CODE (1)
/*
AT END OF DATA

15Systemfunktionen

IF *COUNTER (EMP.) = 0
WRITE 'NO RECORDS FOUND'
ESCAPE BOTTOM

END-IF
WRITE NOTITLE / 'SALARY STATISTICS:'

/ 7X 'MAXIMUM:' MAX(SALARY(1)) CURR-CODE (1)
/ 7X 'MINIMUM:' MIN(SALARY(1)) CURR-CODE (1)
/ 7X 'AVERAGE:' AVER(SALARY(1)) CURR-CODE (1)

END-ENDDATA                                                  
  /*                                                           
END-FIND                                                       
*                                                              
END                                                         ↩

Ausgabe des Programms AEDEX1S:

PERSONNEL         NAME              FIRST-NAME        ANNUAL   CURRENCY
   ID                                                 SALARY     CODE  
--------- -------------------- -------------------- ---------- --------
                                                                       
11100328  BERGHAUS             ROSE                      70800 DM      
11100329  BARTHEL              PETER                     42000 DM      
11300313  AECKERLE             SUSANNE                   55200 DM      
11300316  KANTE                GABRIELE                  61200 DM      
11500304  KLUGE                ELKE                      49200 DM      
                                                                       
SALARY STATISTICS:                                                     
       MAXIMUM:      70800 DM                                          
       MINIMUM:      42000 DM                                          
       AVERAGE:      55680 DM                                         ↩

Beispiel 4 − AT END OF PAGE-Statement mit Systemfunktion AVER

** Example 'AEPEX1S': AT END OF PAGE (structured mode)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 JOB-TITLE
2 SALARY (1)
2 CURR-CODE (1)

END-DEFINE
*
FORMAT PS=10
LIMIT 10
READ EMPLOY-VIEW BY PERSONNEL-ID FROM '20017000'

DISPLAY NOTITLE GIVE SYSTEM FUNCTIONS
NAME JOB-TITLE 'SALARY' SALARY(1) CURR-CODE (1)

/*
AT END OF PAGE

WRITE / 28T 'AVERAGE SALARY: ...' AVER(SALARY(1)) CURR-CODE (1)
END-ENDPAGE

END-READ     
*            
END      ↩

Ausgabe des Programms AEPEX1S:

NAME                  CURRENT            SALARY   CURRENCY
                             POSITION                       CODE  
-------------------- ------------------------- ---------- --------
                                                                  
CREMER               ANALYST                        34000 USD     
MARKUSH              TRAINEE                        22000 USD     
GEE                  MANAGER                        39500 USD     
KUNEY                DBA                            40200 USD     
NEEDHAM              PROGRAMMER                     32500 USD     
JACKSON              PROGRAMMER                     33000 USD     
                                                                  
                           AVERAGE SALARY: ...      33533 USD    ↩

17Systemfunktionen

3 Mathematische Systemfunktionen

In logischen Bedingungen und den Arithmetik-Statements ADD, COMPUTE, DIVIDE, MULTIPLY und
SUBTRACT können Sie die folgenden mathematischen Funktionen verwenden:

Ausgegebener WertFormat/LängeFunktion

Absoluter Wert eines Feldes.wie Feld (field)ABS(field)

Arcustangens eines Feldes.F8ATN(field)

Kosinus eines Feldes.F8COS(field)

Ist der Wert des Feldes größer oder gleich 1017, ist COS(field) "1".

Potenz eines Feldes (e field) mit der Basis e, wobei e die Basis
natürlicher Logarithmen ist.

F8EXP(field)

Bruchteil (hinter dem Komma) eines Feldes.wie Feld (field)FRAC(field)

Ganzzahliger Teil eines Feldes (Integer).wie Feld (field)INT(field)

Natürlicher Logarithmus eines Feldes.

Ein negativer Wert im Argumentfeld wird als positiver Wert
behandelt.

F8LOG(field)

Vorzeichen (Sign) eines Feldes (−1, 0, +1).wie Feld (field)SGN(field)

Sinus eines Feldes.F8SIN(field)

Ist der Wert des Feldes größer oder gleich 1017, ist SIN(field) "0".

Quadratwurzel eines Feldes (Square Root).(*)SQRT(field)

Ein negativer Feldwert wird wie ein positiver behandelt.

Tangens eines Feldes.F8TAN(field)

Ist der Wert des Feldes größer oder gleich 1017, ist TAN(field) "0".

Numerischer Wert eines alphanumerischen Feldes. Der Wert des
Feldesmuss ein in alphanumerischer Form (Codepage oderUnicode)

wie Zielfeld (field)VAL(field)

dargestellter numerischer Wert sein. Leerstellen vor und nach dem

Ausgegebener WertFormat/LängeFunktion

Komma im Feld werden ignoriert. Komma (Dezimalpunkt) und
Vorzeichen werden mitverarbeitet.

Wenn das Zielfeld nicht lang genug ist, werdenNachkommastellen
abgeschnitten (vgl. Abschneiden und Runden von Feldwerten im
Abschnitt Regeln für arithmetische Operationen im Leitfaden zur
Programmierung).

*Diese Funktion wird wie folgt ausgewertet:

■ Wenn das Feld Format/Länge F4 hat, hat SQRT(field) auch Format/Länge F4.
■ Wenn das Feld Format/Länge F8 oder I hat, hat SQRT(field) Format/Länge F8.
■ Wenn das Feld Format N oder P hat, siehe Informationen zu Format/Länge von SQRT(field)
imAbschnittGenauigkeit der Ergebnisse bei arithmetischenOperationen im Leitfaden zur Programmie-
rung.

hat SQRT(field) Format/Länge Nn.7 bzw. Pn.7 (wobei n automatisch groß genug berechnet
wird).

Ein mit einer mathematischen Funktion (außer VAL) verwendetes Feld kann eine Konstante oder
ein Skalar sein undmuss numerisches, gepackt numerisches, Ganzzahl- oderGleitkomma-Format
(N, P, I oder F) haben.

Ein mit der Funktion VAL verwendetes Feld kann eine Konstante, ein Skalar oder ein Array sein
und muss alphanumerisches Format haben.

Beispiel für mathematische Funktionen:

** Example 'MATHEX': Mathematical functions
************************************************************************
DEFINE DATA LOCAL
1 #A (N2.1) INIT <10>
1 #B (N2.1) INIT <-6.3>
1 #C (N2.1) INIT <0>
1 #LOGA (N2.6)
1 #SQRTA (N2.6)
1 #TANA (N2.6)
1 #ABS (N2.1)
1 #FRAC (N2.1)
1 #INT (N2.1)
1 #SGN (N1)
END-DEFINE
*
COMPUTE #LOGA = LOG(#A)
WRITE NOTITLE '=' #A 5X 'LOG' 40T #LOGA
*
COMPUTE #SQRTA = SQRT(#A)

WRITE '=' #A 5X 'SQUARE ROOT' 40T #SQRTA
*
COMPUTE #TANA = TAN(#A)
WRITE '=' #A 5X 'TANGENT' 40T #TANA
*
COMPUTE #ABS = ABS(#B)
WRITE // '=' #B 5X 'ABSOLUTE' 40T #ABS
*
COMPUTE #FRAC = FRAC(#B)
WRITE '=' #B 5X 'FRACTIONAL' 40T #FRAC
*
COMPUTE #INT = INT(#B)
WRITE '=' #B 5X 'INTEGER' 40T #INT
*
COMPUTE #SGN = SGN(#A)
WRITE // '=' #A 5X 'SIGN' 40T #SGN
*
COMPUTE #SGN = SGN(#B)
WRITE '=' #B 5X 'SIGN' 40T #SGN
*
COMPUTE #SGN = SGN(#C)                      
WRITE         '=' #C 5X 'SIGN'        40T #SGN
*                                             
END                                        ↩

Ausgabe des Programms MATHEX:

#A:  10.0     LOG                        2.302585
#A:  10.0     SQUARE ROOT                3.162277
#A:  10.0     TANGENT                    0.648360
                                                 
                                                 
#B:  -6.3     ABSOLUTE                   6.3     
#B:  -6.3     FRACTIONAL                -0.3     
#B:  -6.3     INTEGER                   -6.0     
                                                 
                                                 
#A:  10.0     SIGN                      1        
#B:  -6.3     SIGN                     -1        
#C:   0.0     SIGN                      0       ↩

II Verschiedene Systemfunktionen

Folgende Themen werden behandelt:

*MINVAL/*MAXVAL - Minimum/Maximum auswerten

*TRANSLATE - Zeichen in Groß-/Kleinschreibung umsetzen

*TRIM - Führende und/oder nachfolgende Leerstellen entfernen

POS - Feldidentifikationsfunktion

RET - Returncode-Funktion

SORTKEY - Sortierschlüssel-Funktion

23



24



4 *MINVAL/*MAXVAL - Minimal-/Maximalwertes eines Feldes

■ Funktion ........................................................................................................................................ 26
■ Einschränkungen ............................................................................................................................ 26
■ Syntax-Beschreibung ....................................................................................................................... 26
■ Tabelle für die Konvertierung der resulting-format/length-Werte ................................................................ 28
■ Berechnung der result-format-length ................................................................................................... 30

25



([(IR=result-format/length)] operand,...)
*MINVAL

*MAXVAL

format/length: Format und Länge können entweder explizit mit der IR-Klausel angegeben oder
automatisch mit Hilfe den unten aufgeführten Tabellen errechnet werden, siehe Tabelle für die
Konvertierung der resulting-format/length-Werte.

Funktion

Die Natural-Systemfunktion *MINVAL bzw. *MAXVALwertet die Minimal-/Maximalwerte aller
gegebenen Operanden aus. Das Ergebnis ist immer ein Skalarwert. Wird ein Array als Operand
angegeben, dann wird der Minimal- bzw. Maximalwert aller Array-Felder ausgewertet.

Wenn bei Verwendung von alphanumerischen oder binären Daten dieselben Daten als Argument
angegeben sind, (z.B. *MINVAL('AB','AB')), dann ist das Ergebnis das Argument mit dem
kleinsten/größten Längenwert.

Einschränkungen

Für die Verwendung der Systemfunktion *MINVAL/*MAXVAL gelten folgende Einschränkungen:

■ *MINVAL/*MAXVAL darf nicht an Stellen verwendet werden, an denen eine Zielvariable erwartet
wird.

■ *MINVAL/*MAXVAL darf nicht in einer anderen Systemfunktion verschachtelt werden

Syntax-Beschreibung

Operanden-Definitionstabelle:

Dynamic DefinitionReferencing
Permitted

Possible FormatsPossible StructureOperand

noyesTDBFIPNUAGASCoperand

Syntax-Elementbeschreibung:

BeschreibungSchlüsselwort

Berechnet den Mininalwert aller gegebenen Operandenwerte.*MINVAL

Berechnet den Mininalwert aller gegebenen Operandenwerte.*MAXVAL

Der Operand bzw. die Operanden, deren Minimal-/Maximalwert(e) von der
Systemfunktion *MINVAL bzw. *MAXVAL errechnet werden soll.

operand

IR-Klausel zur expliziten Angabeder resultierenden Format/Länge-Werte.
Siehe unten.

result-format-length

IR-Klausel

Diese Klausel kann zur expliziten Angabe der result-format/length-Werte für die gesamte
Systemfunktion *MINVAL bzw. *MAXVAL verwendet werden.

IR=result-format/length

format-length

IR=
) DYNAMIC(

A
U
B

Eine Aufstellung der gültigen result-format/length-Werte ist im Abschnitt Tabelle für die
Konvertierung der resulting-format/length-Werte enthalten.

Syntax-Elementbeschreibung:

BeschreibungSchlüsselwort

Der Compiler versucht, die resultierenden Format-/Längenwerte der Gesamtfunktion
zu bestimmen. Wenn dies nicht ohne Einbußen an Genauigkeit möglich ist, muss der

format-length

Programmierer die format-length-Werte mit Hilfe der IR-Operandenerweiterung
angeben.

Format: Alphanumerisch oder binär für dynamische Varible.A, U oder B

Anstelle von festen Format-/Längenwerten kann auch ein alphanumerisches, Unicode-
oder binäres Format mit dynamischer Länge angegeben werden.

DYNAMIC

Beispiel:

27Systemfunktionen

*MINVAL/*MAXVAL - Minimal-/Maximalwertes eines Feldes



DEFINE DATA LOCAL
1 #RESULTI       (I4)
1 #RESULTA       (A20)
1 #RESULTADYN    (A) DYNAMIC
1 #A(I4)         CONST <1234>
1 #B(A20)        CONST <H'30313233'> /* '0123' stored
1 #C(I2/1:3)     CONST <2000, 2100, 2200>
END-DEFINE
*
#RESULTA     := *MAXVAL((IR=A20)         #A, #B)     /*no error, I4->A20 is allowed!
#RESULTADYN  := *MAXVAL((IR=(A)DYNAMIC)  #A, #B)     /*result is (A) dynamic
/* #RESULTI  := *MAXVAL((IR=I4)          #A, #B)     /*compiler error, because conv. ↩
A20->I4 is not allowed!  
#RESULTI     := *MAXVAL((IR=I4)          #A, #C(*))  /*maximum of the array is ↩
evaluated
DISPLAY #RESULTA #RESULTADYN (AL=10) #RESULTI
END

Tabelle für die Konvertierung der resulting-format/length-Werte

Es gibt zwei Arten, die resultierenden Format-/Längenwerte für die gesamte Systemfunktion
*MINVAL/*MAXVAL anzugeben.

■ Explizite Angabe der resultierenden Format-/Längenwerte
■ Implizite Angabe der resultierenden Format-/Längenwerte

Explizite Angabe der resultierenden Format-/Längenwerte

Die resultierenden Format-/Längenwerte für die gesamte Systemfunktion *MINVAL bzw. *MAXVAL
können mit der IR-Klausel angegeben werden. Alle angegebenen Operanden werden in die mit
dieser Klausel festgelegten resultierenden Format-/Längenwerte umgewandelt, wenn dies ohne
Einbußen bezüglich derGenauigkeitmöglich ist. Anschließend erfolgt die BerechnungderMinimal-
bzw. Maximalwerte aller umgewandelten Operanden, und es wird ein einziger Skalarwert mit
dem errechneten Format-/Längenwert als Ergebnis für die gesamte Systemfunktion festgelegt.

Implizite Angabe der resultierenden Format-/Längenwerte

Wenn keine IR-Klausel in der Systemfunktion *MINVAL bzw. *MAXVAL verwendet wird, erfolgt die
Berechnung der resultierenden Format-/Längenwerte anhand der Formate/Längen aller in der
Systemfunktion als Argument angegebenen Operanden. Dabei wird der format/length-Wert
jedes einzelnen Operanden genommen und mit dem format/length-Wert des darauf folgenden
Operanden in der Argument-Liste zusammengefasst. Der resultierenden Format-/Längenwert
von zwei einzelnen Operanden wird dann gemäß der unten aufgeführten Tabellen errechnet.

Die Tabelle für die Konvertierung der resulting-format/length-Werte ist in zwei Untertabellen
aufgeteilt. Kombinationen, die nicht in diesen beiden Tabellen aufgeführt sind, sind ungültig und

dürfen in derArgument-Liste der Systemfunktion *MINVAL bzw. *MAXVAL nicht verwendetwerden.
Das Schlüsselwort FLF gibt an, wann zur Vermeidung von Ungenauigkeiten die IR-Klausel zur
Festlegung des resultierenden Format-/Längenwertes benutzt werden muss.

Tabelle 1

Enthält alle nummerischen Kombinationsmöglichkeiten für zwei verschiedene Operanden.

Zweiter Operand
F4, F8Pa.b, Na.bI4I2I1Format/Länge

F8Pmax(3,a).bI4I2I1I1Erster
Operand F8Pmax(5,a).bI4I2I2I2

F8Pmax(10,a).bI4I4I4I4

Pmax(10,x).yPmax(5,x).yPmax(3,x).yPx.y, Nx.y wenn y=0 und
x <=15;

wennmax(x,a) +max(y,b)
<= 29

F8Pmax(x,a).max(y,b)
sonst FLFsonst FLF

F8F8F8F8F4, F8 wenn b=0 und a <=15
F8

sonst FLF

Legende:

Format-/Längenangabe mittels IR-Klausel zwingend erforderlich.FLF

Format/Länge ist Integer (Ganzzahl). x gibt dieAnzahl der Bytes an, die zumSpeichern
des Integer-Wertes benutzt werden.

Ix

Format/Länge ist Float (Gleitkomma).x gibt dieAnzahl der Bytes an, die zumSpeichern
des Float-Wertes benutzt werden.

Fx

Format ist Packed (gepackt) mit entsprechender Anzahl an Stellen vor dem
Dezimalpunkt (x,a) und der Genauigkeit (y,b).

Px.y

Pa,b

Format ist Numeric (nummerisch) mit entsprechender Anzahl an Stellen vor dem
Dezimalpunkt (x,a) und der Genauigkeit (y,b).

Nx.y
Na,b

Das resultierende Format ist Packed (gepackt). Die Berechnung der Länge erfolgt
anhandder nachfolgenden Informationen.DieAnzahl an Stellen vordemDezimalpunkt
ist der Maximalwert von c und d. Der Genauigkeitswert ist e.

Pmax(c,d).e

Das resultierende Format ist Packed (gepackt). Die Berechnung der Länge erfolgt
anhandder nachfolgenden Informationen.DieAnzahl an Stellen vordemDezimalpunkt

Pmax(c,d).max(e,f)

ist der Maximalwert von c und d. Der Genauigkeitswert ist der Maximalwert von e
und f.

Tabelle 2

Enthält alle Formate und Längen, die für Operanden in den Systemfunktionen *MINVAL/*MAXVAL
verwendet werden können.

Zweiter Operand

Ua, U dynamicBa, B dynamicAa, A dynamicTDFormat-length

NANANATDDErster
Operand NANANATTT

U dynamicA dynamicA dynamicNANAAx, A dynamic

U dynamicB dynamicA dynamicNANABx, B dynamic

U dynamicU dynamicU dynamicNANAUx, U dynamic

Legende:

Unzulässige Kombination.NA

Datenformat.D

Zeitformat.T

Binärformat mit Länge x, a.Bx, Ba

Alphanumerisches Format mit Länge x, a.Ax, Aa

Unicode-Format mit Länge x, a.Ux, Ua

Binärformat mit dynamischer Länge.B dynamic

Alphanumerisches Format mit dynamischer Länge.A dynamic

Unicode-Format mit dynamischer Länge.U dynamic

Berechnung der result-format-length

Anhand der oben aufgeführten Regeln ist der Compiler in der Lage, die Source-Operanden unter
Berücksichtigung von Operandenpaaren zu verarbeiten und die Zwischenergebnisse für jedes
Paar zu berechnen. Das erste Paar besteht aus dem ersten und dem zweiten Operand, das zweite
Paar aus demZwischenergebnis unddemdrittenOperandusw.NachVerarbeitung allerOperanden
stellt das letzte Ergebnis den Vergleich zwischen Format und Länge dar, die zum Vergleich mit
allen Operanden zwecks Berechnung des Minimal- bzw. Maximalwertes verwendet werden. Bei
Verwendung dieser Format-/Längenberechnungsmethode können die Operanden in beliebiger
Reihenfolge auftreten.

Beispiel:

DEFINE DATA LOCAL
1 A (I2) INIT <34>
1 B (P4.2) INIT <1234.56>
1 C (N4.4) INIT <12.6789>
1 D (I1) INIT <100>
1 E (I4/1:3) INIT <32, 6745, 456>
1 #RES-MIN (P10.7)
1 #RES-MAX (P10.7)
END-DEFINE
*
MOVE *MINVAL(A, B, C, D, E(*)) TO #RES-MIN
MOVE *MAXVAL(A, B, C, D, E(*)) TO #RES-MAX
DISPLAY #RES-MIN #RES-MAX
END

Ausgabe:

#RES-MIN #RES-MAX
--------------------------- ---------------------------

12.6789000 6745.0000000

Die folgende Tabelle zeigt die einzelnen Schritte zur automatischen Format-/Längenberechnung
im obigen Beispiel. Sie enthält die Zwischenergebnisse (ir) aller Schritte und den Format-/Längen-
wert des Vergleichs (cf), der als result-format/length-Wert verwendet wird.

Format/Länge des
Zwischenergebnisses(ir)

Format/Länge des
zweiten Operanden
bzw. Zwischenergebnis
(ir)

Name des
zweiten
Operanden

Format/Länge des
erstenOperanden bzw.
Zwischenergebnis (ir)

Name des
ersten
Operanden

Berechnungs-
reihenfolge

ir1 = P5.2P4.2BI2A1.

ir2 = P5.4N4.4CP5.2ir12.

ir3 = P5.4I1DP5.4ir23.

cf = P10.4I4EP5.4ir34.

Zur Laufzeit werden alle Operanden in die cf-Format/Länge umgewandelt. Danach werden alle
umgewandeltenWertemiteinander verglichenundder entsprechendeMinimal- bzw.Maximalwert
wird errechnet.

Anmerkungen:

1. Wenn nur ein einzelner Operand angegeben wird, hat result-format-length der Format-
/Längenwert dieses Operanden.

2. Wenn ein binärer Operand mit einer Länge im Bereich von 1 - 4 als Argument innerhalb der
Systemfunktion *MINVAL/*MAXVAL zusammenmit einemalphanumerischen oder einemUnicode-

Operanden angegeben wird, dann wird das Zwischenergebnis (result-format-length) als
alphanumerisches bzw. als Unicode-Format mit dynamischer Länge bewertet.

In diesem Fall wird der Wert des binären Operanden als numerischer Wert betrachtet, der
gemäß den Datenübertragungsregeln in das result-format-length umgewandelt wird (der
binäre numerische Wert wird in das Format Upacked gewandelt), bevor der Minimal-/Maxi-
malwert berechnet wird.

Beispiel:

DEFINE DATA LOCAL
1 #B4 (B4) INIT <1>
1 #A10(A10) INIT <"2">
END-DEFINE
WRITE "=" *MAXVAL(#A10, #B4) (AL=60) /* RESULT FORMAT-LENGTH IS (A)DYNAMIC: "2"
WRITE "=" *MINVAL(#A10, #B4) (AL=60) /* RESULT FORMAT-LENGTH IS (A)DYNAMIC: "1"
END

Der Zwischenwert result-format-length (#A10, #B4) ist A dynamisch.

Daher wird zunächst #A10 nach A dynamisch gewandelt, ebensowie #B4 nach A dynamisch
gewandelt wird (unter Berücksichtigung der Datenübertragungsregeln), bevor das Zwischen-
ergebnis der beiden Operanden berechnet wird.

Reihenfolge der Berechnung von Format und Länge

Die folgende Grafik zeigt die Reihenfolge, in der die Berechnung von Format und Länge erfolgt:

Legende:

Zwischenergebnis (intermediate result) 1, 2, 3.ir1, ir2, ir3

Resultierende Vergleich-Format- und Längewerte (comparison).cf

),*TRANSLATE (operand
LOWER
UPPER

Format/Länge: wie bei operand.

Funktion

Die Systemfunktion *TRANSLATE dient zumUmsetzen vonZeichen einesOperandenmit Alphanu-
merischem oder binärem Format in Groß- oder Kleinbuchstaben. Der Inhalt des Operanden bleibt
dabei unverändert.

*TRANSLATE kann als operand an jeder Stelle eines Statements angegeben werden, an der ein
Operand mit Format A oder B zulässig ist.

Einschränkungen

Für die Verwendung der Systemfunktion *TRANSLATE gelten folgende Einschränkungen:

■ *TRANSLATE darf nicht an Stellen verwendet werden, an denen eine Zielvariable erwartet wird.
■ *TRANSLATE darf nicht in einer anderen Systemfunktion verschachtelt werden.

Syntax-Beschreibung

Operanden-Definitionstabelle:

Dynam. DefinitionReferenzierung
erlaubt

Mögliche FormateMögliche StrukturOperand

neinjaBUAASCoperand

Syntax-Elementbeschreibung:

Umsetzung in Kleinbuchstaben*TRANSLATE (operand,LOWER)
Bei Angabe des Schlüsselworts LOWER als zweites Argument angegeben
wird die Zeichenkette in operand in Kleinbuchstaben umgesetzt.

Umsetzung in Großbuchstaben*TRANSLATE (operand,UPPER)
Bei Angabe des Schlüsselworts UPPER als zweites Argument angegeben
wird die Zeichenkette in operand in Großbuchstaben umgesetzt.

Beispiel

DEFINE DATA LOCAL
1 #SRC (A)DYNAMIC INIT <'aBcDeFg !§$%&/()=?'>
1 #DEST (A)DYNAMIC
END-DEFINE
*
PRINT 'Source string to be translated:..........' #SRC
*
MOVE *TRANSLATE(#SRC, UPPER) TO #DEST
PRINT 'Source string translated into upper case:' #DEST
*
MOVE *TRANSLATE(#SRC, LOWER) TO #DEST
PRINT 'Source string translated into lower case:' #DEST
END

Ausgabe:

Source string to be translated:.......... aBcDeFg !§$%&/()=?

Source string translated into upper case: ABCDEFG !§$%&/()=?

Source string translated into lower case: abcdefg !§$%&/()=?

),*TRIM (operand
LEADING

TRAILING

Format/Länge: wie bei operand (A oder B)/DYNAMIC.

Funktion

Die Systemfunktion *TRIMdient zumEntfernen von führendenund/oder nachfolgendenLeerstellen
aus einer aphanumerischen oder binären Zeichenkette. Der Inhalt von operand bleibt dabei
unverändert. Bei Verwendung einer dynamischen Variablen als operand, wird die Länge dieser
Variablen dem Ergebnis entsprechend angepasst.

Die Systemfunktion *TRIM kann als operand an jeder Stelle eines Statements angegeben werden,
an der ein Operand mit Format A oder B zulässig ist.

Einschränkungen

Für die Verwendung der Systemfunktion *TRIM gelten folgende Einschränkungen:

■ *TRIM darf nicht an Stellen verwendet werden, an denen eine Zielvariable erwartet wird.
■ *TRIM darf nicht in einer anderen Systemfunktion verschachtelt werden.
■ Wenn der Operand eine statische Variable ist, kann manmit *TRIM keine führenden Leerstellen
entfernen, weil bei statischen Variablen die verbleibenden nachfolgenden Stellen des Variablen-
speichers mit Leerzeichen aufgefüllt werden.

Syntax-Beschreibung

Operanden-Definitionstabelle:

Dynam. DefinitionReferenzierung
erlaubt

Mögliche FormateMögliche StrukturOperand

neinjaBUAASCoperand

Syntax-Elementbeschreibung:

Führende Leerstellen entfernen*TRIM(operand,
LEADING)

Bei Angabe des Schlüsselworts LEADING als zweites Argument werden alle
führenden Leerstellen aus dem in operand enthaltenen String entfernt.

Nachfolgende Leerstellen entfernen*TRIM(operand,
TRAILING)

Bei Angabe des Schlüsselworts TRAILING als zweites Argument werden alle
nachfolgenden Leerstellen aus dem in operand enthaltenen String entfernt.

Führende und Nachfolgende Leerstellen entfernen*TRIM(operand)

Wennkein zweites Schlüsselwort alsArgument angegebenwird, bewirktTRIM,
dass alle führenden und nachfolgenden Leerstellen aus dem in operand
enthaltenen String entfernt werden.

Beispiel

Verwendung eines alphanumerischen Arguments

DEFINE DATA LOCAL
/*******************************
/* STATIC VARIABLE DEFINITIONS
/*******************************
1 #SRC (A15) INIT <' ab CD '>
1 #DEST (A15)

/* FOR PRINT OUT WITH DELIMITERS
1 #SRC-PRN (A20)
1 #DEST-PRN (A20)

/*******************************
/* DYNAMIC VARIABLE DEFINITIONS
/*******************************
1 #DYN-SRC (A)DYNAMIC INIT <' ab CD '>
1 #DYN-DEST (A)DYNAMIC

/* FOR PRINT OUT WITH DELIMITERS
1 #DYN-SRC-PRN (A)DYNAMIC
1 #DYN-DEST-PRN (A)DYNAMIC

END-DEFINE

PRINT 'static variable definition:'
PRINT '---------------------------'
COMPRESS FULL ':' #SRC ':' TO #SRC-PRN LEAVING NO SPACE
PRINT ' '
PRINT ' 123456789012345 123456789012345'

MOVE *TRIM(#SRC, LEADING) TO #DEST

COMPRESS FULL ':' #DEST ':' TO #DEST-PRN LEAVING NO SPACE
DISPLAY #SRC-PRN #DEST-PRN '*TRIM(#SRC, LEADING)'

MOVE *TRIM(#SRC, TRAILING) TO #DEST
COMPRESS FULL ':' #DEST ':' TO #DEST-PRN LEAVING NO SPACE
DISPLAY #SRC-PRN #DEST-PRN '*TRIM(#SRC, TRAILING)'

MOVE *TRIM(#SRC) TO #DEST
COMPRESS FULL ':' #DEST ':' TO #DEST-PRN LEAVING NO SPACE
DISPLAY #SRC-PRN #DEST-PRN '*TRIM(#SRC)'

PRINT ' '
PRINT 'dynamic variable definition:'
PRINT '----------------------------'
COMPRESS FULL ':' #DYN-SRC ':' TO #DYN-SRC-PRN LEAVING NO SPACE
PRINT ' '
PRINT ' 1234567890 12345678'

MOVE *TRIM(#DYN-SRC, LEADING) TO #DYN-DEST
COMPRESS FULL ':' #DYN-DEST ':' TO #DYN-DEST-PRN LEAVING NO SPACE
DISPLAY (AL=20) #DYN-SRC-PRN #DYN-DEST-PRN '*TRIM(#SRC, LEADING)'

MOVE *TRIM(#DYN-SRC, TRAILING) TO #DYN-DEST
COMPRESS FULL ':' #DYN-DEST ':' TO #DYN-DEST-PRN LEAVING NO SPACE
DISPLAY (AL=20) #DYN-SRC-PRN #DYN-DEST-PRN '*TRIM(#SRC, TRAILING)'

MOVE *TRIM(#DYN-SRC) TO #DYN-DEST
COMPRESS FULL ':' #DYN-DEST ':' TO #DYN-DEST-PRN LEAVING NO SPACE
DISPLAY (AL=20) #DYN-SRC-PRN #DYN-DEST-PRN '*TRIM(#SRC)'

PRINT ' '
PRINT '":" := delimiter character to show the start and ending of a string!'
END

Ausgabe:

#SRC-PRN #DEST-PRN
-------------------- -----------------------

static variable definition:
----------------------------

123456789012345 123456789012345
: ab CD : :ab CD : *TRIM(#SRC, LEADING)
: ab CD : : ab CD : *TRIM(#SRC, TRAILING)
: ab CD : :ab CD : *TRIM(#SRC)

dynamic variable definition:
----------------------------

1234567890 12345678
: ab CD : :ab CD : *TRIM(#SRC, LEADING)

: ab CD : : ab CD: *TRIM(#SRC, TRAILING)
: ab CD : :ab CD: *TRIM(#SRC)

':' := delimiter character to show the start and ending of a string!

43Systemfunktionen

*TRIM - Entfernen von führenden und/oder nachfolgenden Leerstellen



44



7 POS - Feldidentifikationsfunktion

I4Format/Länge:

Die Systemfunktion POS(field-name) enthält die interne Identifikation des Feldes, dessen Name
mit der Systemfunktion angegeben wird.

POS(field-name) identifiziert ein bestimmtes Feld, unabhängig von seiner Position in einerMaske
(Map). Auchwenn sich die Reihenfolge undAnzahl der Felder in einerMaske ändert, identifiziert
POS(field-name) nach wie vor eindeutig dasselbe Feld. Damit genügt zum Beispiel ein einziges
REINPUT-Statement, um es von der Programmlogik abhängig zu machen, welches Feld MARKiert
werden soll.

Beispiel:

DECIDE ON FIRST VALUE OF ...
VALUE ...

COMPUTE #FIELDX = POS(FIELD1)
VALUE ...

COMPUTE #FIELDX = POS(FIELD2)
...

END-DECIDE
...
REINPUT ... MARK #FIELDX

Wenn das mit POS angegebene Feld ein Array ist, muss eine bestimmte Ausprägung angegeben
werden; zum Beispiel POS(FIELDX(5)). Auf einen Array-Bereich kann POS nicht angewendet
werden.

POS und *CURS-FIELD

POS(field-name) kann in Verbindung mit der Natural-Systemvariablen *CURS-FIELD dazu ver-
wendet werden, die Ausführung bestimmter Funktionen davon abhängig zumachen, inwelchem
Feld der Cursor zur Zeit steht.

*CURS-FIELD enthält die interne Identifikation des Feldes, in dem sich der Cursor zur Zeit befindet;
*CURS−FIELDkannnicht alleine, sondernnur zusammenmit POS(field-name)verwendetwerden.
Sie können beide zusammen dazu benutzen, um zu prüfen, ob sich der Cursor gerade in einem
bestimmten Feld befindet, und die weitere Verarbeitung von dieser Bedingung abhängigmachen.

Beispiel:

IF *CURS-FIELD = POS(FIELDX)
MOVE *CURS-FIELD TO #FIELDY

END-IF
...
REINPUT ... MARK #FIELDY

Anmerkungen:

1. Die Werte von *CURS-FIELD und POS(field-name) dienen nur zur internen Identifikation der
Felder und können nicht für arithmetische Operationen verwendet werden.

2. Der von POS(field-name) zurückgegebene Wert für eine Ausprägung eines X-Arrays (ein
Array, für das wenigstens eine Dimension als erweiterbar angegeben ist) kann sich ändern,
nachdem die Anzahl der Ausprägungen für eine Dimension des Arrays mittels der Statements
EXPAND, RESIZE oder REDUCE geändert wurde.

3. Natural RPC:Wenn *CURS-FIELD und POS(field-name) sich auf eine Kontextvariable beziehen,
können die daraus resultierenden Informationen nur innerhalb derselbenKonversation verwen-
det werden.

4. InNatural forAjax-Anwendungendient *CURS-FIELD zur IdentifikationdesOperanden,welcher
den Wert des Control darstellt, welches den Eingabefokus hat. Sie können *CURS-FIELD in
Verbindung mit der POS-Funktion benutzen, um eine Prüfung auf das Control, welches den
Eingabefokus hat, zu veranlassen und die Verarbeitung in Abhängigkeit von diesem Zustand
durchzuführen.

Siehe auch

■ Dialog-Gestaltung, Feld-sensitive Verarbeitung und Einfachere Programmierung imNatural Leitfaden
zur Programmierung.

■ POS22 - Version 2.2 Algorithm for POS System Function in der Parameter-Referenz-Dokumentation.

8 RET - Returncode-Funktion

I4Format/Länge:

Die Systemfunktion RET(program-name) kann dazu verwendet werden, den Returncode eines
nicht in Natural geschriebenen Programms, das über ein CALL-Statement aufgerufen wurde, zu
erhalten.

RET(program-name) kann in einem IF-Statement sowie in denArithmetik-Statements ADD, COMPUTE,
DIVIDE, MULTIPLY und SUBTRACT verwendet werden.

Beispiel:

DEFINE DATA LOCAL
1 #RETURN (I4)
...
END-DEFINE
...
...
CALL 'PROG1'
IF RET('PROG1') > #RETURN

WRITE 'ERROR OCCURRED IN PROGRAM 1'
END-IF
...

9 SORTKEY - Sort-Key Function

SORTKEY (character-string)

Diese Systemfunktion dient zum Konvertieren von „nicht richtig sortierten“ Zeichen (oder Kom-
binationen von Zeichen) in andere Zeichen (oder Kombinationen von Zeichen), die vom Sortier-
programm oder Datenbanksystem alphabetisch „richtig sortiert“ werden.

A253Format/Länge:

Es gibt in vielen LandessprachenZeichen (oder Zeichenkombinationen), die von einem Sortierpro-
gramm oder Datenbanksystem nicht in der richtigen alphabetischen Reihenfolge sortiert werden,
da die Reihenfolge der Zeichen im vom Computer verwendeten Zeichensatz nicht immer der
alphabetischen Reihenfolge der Zeichen entspricht.

Zum Beispiel wird der spanische Buchstabe CH in der Regel von einem Sortierprogramm bzw.
Datenbanksystem wie zwei Buchstaben behandelt und zwischen CG und CI einsortiert — gehört
aber eigentlich als eigener Buchstabe im spanischen Alphabet zwischen "C" und "D".

Oder es kann sein, dass Kleinbuchstaben und Großbuchstaben entgegen IhrenWünschen bei der
Sortierreihenfolge nicht gleich behandelt werden, dass Ziffern vor Buchstaben sortiert werden
(Sie aber wünschen, dass Buchstaben vor Ziffern sortiert werden) oder dass Sonderzeichen (z.B.
Bindestriche in Doppelnamen) zu einer unerwünschten Sortierreihenfolge führen.

In solchen Fällen können Sie die Systemfunktion SORTKEY(character-string) benutzen. Die von
SORTKEY berechnetenWertewerden nur als Sortierkriteriumbenutzt, während die ursprünglichen
Werte für die Interaktion mit dem Endbenutzer verwendet werden.

Sie können die SORTKEY-Funktion in einem COMPUTE sowie in einer logischen Bedingung als arith-
metischen Operanden verwenden.

Als character-string können Sie eine alphanumerische Konstante oder Variable oder eine ein-
zelne Ausprägung eines alphanumerischen Arrays angeben.

Wenn Sie die SORTKEY-Funktion in einem Natural-Programm angeben, wird der User-Exit
NATUSKnn aufgerufen — wobei nn der aktuelle Sprachcode (d.h. der aktuelle Wert der System-
variablen *LANGUAGE) ist.

Sie können diesen User-Exit in jeder Programmiersprache, die über eine Standard-CALL-Schnitt-
stelle verfügt, schreiben. Der mit SORTKEY angegebene character-stringwird an den User-Exit
übergeben. Der User-Exit muss so programmiert werden, dass er „falsch sortierte“ Zeichen in
dieser Zeichenkette in entsprechende „richtig sortierte“ Zeichen umsetzt. Die umgesetzte Zeichen-
kette wird dann vom Natural-Programm zur weiteren Verarbeitung verwendet.

Allgemeine Aufruf-Konventionen für externe Programme sind in der Dokumentation zum
CALL-Statement erläutert.

Nähere Informationen zudenAufruf-Konventionen für SORTKEYUser-Exits finden Sie imAbschnitt
User Exit for Computation of Sort Keys in der Natural Operations Documentation.

Beispiel:

DEFINE DATA LOCAL
1 CUST VIEW OF CUSTOMERFILE

2 NAME
2 SORTNAME

END-DEFINE
...
*LANGUAGE := 4
...
REPEAT

INPUT NAME
SORTNAME := SORTKEY(NAME)
STORE CUST
END TRANSACTION
...

END-REPEAT
...
READ CUST BY SORTNAME

DISPLAY NAME
END-READ
...

Angenommen, im obigen Beispiel würden bei mehrmaliger Ausführung des INPUT-Statements
nacheinander die Werte "Sanchez", "Sandino" und "Sancinto" eingegeben.

Bei der Zuweisung von SORTKEY(NAME) zu SORTNAMEwürde der User-Exit NATUSK04 aufgerufen.
Diesermüsste so programmiertwerden, dass er zunächst alle Kleinbuchstaben inGroßbuchstaben
umsetzt und dann die Zeichenfolge "CH" in "Cx" umsetzt — wobei x dem letzten Zeichen im
verwendeten Zeichensatz entspricht, also hexadezimal H’FF’ (vorausgesetzt dieses letzte Zeichen
ist kein druckbares Zeichen).

Es werden sowohl die „eigentlichen“ Namen (NAME) als auch die für die gewünschte Sortierung
umgesetzten Namen (SORTNAME) gespeichert. Zum Lesen der Datei wird SORTNAME verwendet.
Dann würden bei Ausführung des DISPLAY-Statements die Namen in der richtigen spanischen
alphabetischen Reihenfolge ausgegeben:

Sancinto
Sanchez
Sandino

III Funktionen, die als Natural-Objekte ausgeliefert werden

Dieses Dokument beschreibt Funktionen, die unter Verwendung von Natural-Objekten des Typs
Function implementiert wurden.

Diese Function-Objekte (und ihre Prototyp-Definitionen), derenNamenmit SAG beginnen,werden
in der Natural-Library SYSTEM in der Systemdatei FNAT ausgeliefert. Beispiele für Function Calls
sind in der Library SYSEXPG vorhanden.

Weitere Informationen siehe Function Call im Leitfaden zur Programmierung.

URL-Kodierung

Bei HTTP-Anfragen über Natural-Anwendungsschnittstellen ist es oftmals notwendig, dass der
URI (Uniform Resource Identifier) URL-kodiert ist. Das Statement REQUEST DOCUMENT benötigt
eine solche URL, um auf ein Dokument zugreifen zu können.

URL-Kodierung, auch Prozentkodierung genannt, ist einMechanismus, umbestimmte Sonderzei-
chen in Teilen einer URL zu ersetzen. Zur Bildung einer URL werden nur Zeichen des US-ASCII-
Zeichensatzes verwendet. Einige Zeichen des US-ASCII-Zeichensatzes haben bei Verwendung in
einer URL eine besondere Bedeutung - sie sind als „reservierte“ Steuerzeichen eingestuft, die zur
Strukturierung der Zeichenkette in verschiedene sematische Unterkomponenten dienen. Der
Quasi-Standard bezüglich der generischen Syntax einerURL ist imDokument RFC3986 enthalten,
das von der Internet-Community zusammengestellt worden ist. Darin wird beschrieben, unter
welchen Bedingungen eine URL-Kodierung notwendig ist. Dazu gehört auch die Darstellung von
Zeichen,welche nicht Bestandteil desUS-ASCII-Zeichensatzes sind (zumBeispiel das Euro-Zeichen)
und die Verwendung der reservierten Zeichen.

Reservierte Zeichen sind:

][@;:/,+*)('%$!#&=?

Nicht reservierte Zeichen sind:

~._-0-9a-zA-Z

Eine URL besteht aus reservierten und nicht-reservierten Zeichen; andere Zeichen dürfen in ihr
nicht vorkommen. Wenn andere Byte-Werte benötigt werden (die keinem der reservierten und
nicht reserviertenZeichen entsprechen) oderwenn reservierte Zeichen alsDaten verwendetwerden
(die keine besondere semantische Bedeutung im URL-Kontext haben sollen), müssen dieseWerte
in die „Prozentkodierung“ konvertiert werden: ein Prozentzeichen und direkt danach die aus
zwei Zeichen bestehende hexadezimale Darstellung des Codepoint (bedingt durch das Kodier-
schema Windows-1252). Das hat zur Folge, dass ein Pluszeichen (+) als %2B, ein Prozentzeichen
als %25 und ein „at“-Zeichen (@) als %40 in der Zeichenkette erscheint.

Die nachfolgendbeschriebenenKodierfunktionen bearbeiten die kompletten Eingabe-Zeichenkette.
Bitte beachten Sie, dass Sie keine kompletteURLoder Teile davon kodieren, falls diese Steuerzeichen
(reservierte Zeichen) enthalten, die nicht in die Prozentform umgesetzt werden dürfen. Die hier
beschriebenen Funktionen sollten nur bei Zeichen angewendet werden, welche nicht für den
Gebrauch in einer URL erlaubt sind, und bei Zeichen mit einer besonderen Bedeutung im URL-
Kontext, die als Datenelement übergeben werden.

■ Einfache Kodierung
■ SAGENC - Einfache Kodierung (Format A nach Format A)
■ SAGDEC - Einfache Dekodierung (Format A nach Format A)
■ Erweiterte Kodierung
■ SAGENCE - Erweiterte Kodierung (Format U nach Format A, optionale Parameter)
■ SAGDECE - Erweiterte Dekodierung (Format A nach Format U, optionale Parameter)
■ Beispiel-Programm

Einfache Kodierung

Der einzelne Eingabeparameter enthält die zu kodierende oder zu dekodierende Zeichenkette.
Alle darin enthaltenen Daten werden so betrachtet, als ob sie in der Codepage EBCDIC 1140 dar-
gestellt sind, und zwar unabhängig davonwelche Session-Codepage zurzeit aktiv ist. ZurAusfüh-
rungder Funktion SAGENC bzw. SAGDEC ist keineUnicode-Unterstützung erforderlich.Die folgenden
Zeichen werden durch die entsprechenden hexadezimalen US-ASCII-Zeichen ersetzt.

\}{][^"='@#:`?>%,/;)*$!&|+(<Das
Zeichen,...

E0D0C0BBBAB07F7E7D7C7B7A796F6E6C6B615E5D5C5B5A504F4E4D4Cdas sich
am
EBCDIC
Codepoint
befindet,

5C7D7B5D5B5E223D2740233A603F3E252C2F3B292A2421267C2B283Cwird
kodiert zu
%nn

Folgende Funktionen stehen zur Verfügung:

■ SAGENC - Einfache Kodierung (Format A nach Format A)
■ SAGDEC - Einfache Dekodierung (Format A nach Format A)

SAGENC - Einfache Kodierung (Format A nach Format A)

Die Funktion SAGENC kodiert eine Zeichenkette in die prozentkodierte Form. Gemäß Standard
RFC3986 werden reservierte Zeichen und Zeichen unterhalb von US-ASCII x'7F' (die in einer
URL nicht erlaubt sind) prozentkodiert; ein Leerzeichen wird durch ein Pluszeichen (+) ersetzt.
Nicht reservierte Zeichen gemäßRCF3986 undZeichen oberhalb vonUS-ASCII x'7F', zumBeispiel
deutsche Umlaute, werden nicht kodiert. Falls Sie solche Zeichen kodieren möchten, müssen Sie
die erweitere Kodierfunktion SAGENCE.

BeschreibungObjekt

Dies ist der Aufruf der einfachen Kodierfunktion.SAGENC

Der Copycode, der die Prototyp-Definition enthält, wird nur bei der Kompilierung benutzt, um
den Typ der Rückgabevariablen als Referenz für den Funktionsaufruf zu bestimmen und um
die Parameter zu prüfen, falls dies gewünscht wird.

SAGENCP ist optional.

SAGENCP

Beispiel-Programm in der Library SYSEXPG.URLX01

#URL-ENC := SAGENC(<#URL-DEC>)

SAGDEC - Einfache Dekodierung (Format A nach Format A)

Die Funktion SAGDEC dient zum Dekodieren der von der Funktion SAGENC gelieferten Prozentko-
dierungen. Außer der zu dekodierenden Zeichenkettewerden keineweiteren Parameter benötigt.

BeschreibungObjekt

Dies ist der Aufruf der einfachen Dekodierfunktion.SAGDEC

Der Copycode, der die Prototyp-Definition enthält, wird nur bei der Kompilierung benutzt, um
den Typ der Rückgabevariablen als Referenz für den Funktionsaufruf zu bestimmen und um
die Parameter zu prüfen, falls dies gewünscht wird.

SAGDECP ist optional.

SAGDECP

Beispiel-Programm in der Library SYSEXPG.URLX01

#URL-DEC := SAGDEC(<#URL-ENC>)

Erweiterte Kodierung

Die erweiterte Funktion berücksichtigt alle in der Spezifikation RFC3986 aufgeführten Sonderfälle.
Die folgenden Parameter können berücksichtigt werden (Die Standardvorgabewerte sind fett
hervorgehoben.):

1. <dynamic U-string> soll kodiert/dekodiert werden.

2. Return Code: <> 0 (Natural-Fehler) bei Auftreten eines Fehlers im MOVE ENCODED-Statement.

3. Fehlerzeichen falls Return Code <> 0.

4. Leerzeichen: %20/+/nicht kodieren (Standardeinstellung: +)

5. Nicht reservierte Zeichen: kodieren/nicht kodieren

6. Reservierte Zeichen: kodieren/nicht kodieren

7. Sonstige Sonderzeichen (weder nicht reservierte noch reservierte Zeichen): kodieren/nicht
kodieren

8. Zeichen-Prozentkodierung: ISO-8859-1/UTF-8/beliebige andere Codepage/falls = ' ' dann
*CODEPAGE (Natural-Standard-Codepage, nicht die Codepage für die Standardkodierung!)

9. VomBenutzer gewähltes Zeichen in einemX-array des FormatsU,welches nicht nach den oben
aufgeführten Parametern prozentkodiert werden soll (zum Beispiel das Euro-Symbolzeichen,
das in der Codepage ISO-8859-1 nicht vorhanden ist) oder um zu verhindern, dass ein Zeichen
prozentkodiert wird.

10. Benutzer-definierte Prozentkodierung in ein X-array des Formats A, für ein vom Benutzer
gewähltes Zeichen in derselben Ausprägung des X-array.

Der Eingabeparameter für eine Zeichenkette hat das Natural-Format U. Das bedeutet, dass die
Eingabezeichenkette alleUnicode-Zeichen enthalten darf. DieAusgabezeichenkette der erweiterten
Funktion hat dasNatural-FormatAund ist in derNatural-Standard-Codepage (*CODEPAGE) kodiert.
Die Codepage der Prozentkodierung kann gewählt werden. Die Prozentkodierung gemäß UTF-
8, ISO-8859-1, des Euro-Symbolzeichens erfolgt mittels eines MOVE ENCODED-Statements. Ist ein
Eingabezeichen in der für die Prozentkodierung verwendeten Ziel-Codepage nicht vorhanden,
dann wird dieses Zeichen nicht kodiert. Das bedeutet, das Zeichen wird unverändert in der
Natural-Standard-Codepage zurückgegeben. Falls das Zeichen in derNatural-Standard-Codepage
auch nicht vorhanden ist, wird es durch das vom MOVE ENCODED-Statement zurückgelieferte
Ersetzungszeichen ersetzt. Das Ersetzungszeichenwird prozentkodiert. Dieser Fall kann nur dann
eintreten, wenn die Codepage zur Prozentkodierung nicht UTF-8 ist. Der zuletzt aufgetretene
MOVE ENCODED-Fehler wird zurückgeliefert.

Die Parameter sind optional.Wenn der Benutzer einen Parameter nicht angibt, wird der Standard-
vorgabewert verwendet. Falls der Benutzer eine eigene Zeichenumsetzungstabelle angibt, werden
die Zeichen in der Tabelle entsprechend dieser Tabelle und nicht entsprechend den anderen
Parametern prozentkodiert.Wenn die Prozentkodierung eines Zeichens in der benutzerdefinierten
Umsetzungstabelle gleich dem Zeichen oder leer ist, wird dieses Zeichen nicht kodiert. Somit

können einzelne Zeichen aus dem reservierten oder nicht reservierten Zeichensatz ausgeschlossen
werden.

Folgende Funktionen stehen zur Verfügung:

■ SAGENCE - Erweiterte Kodierung (Format U nach Format A, optionale Parameter)
■ SAGDECE - Erweiterte Dekodierung (Format A nach Format U, optionale Parameter)

SAGENCE - Erweiterte Kodierung (Format U nach Format A, optionale Parameter)

Die Funktion SAGENCE dient zur Prozentkodierung einer Zeichenkette unter Verwendung des
Hexadezimalwerts aus der gewählten Codepage (standardmäßi UTF-8). Gemäß dem Standard
RFC3986 werden reservierte Zeichen und Zeichen unterhalb von US-ASCII x'7F', welche in
einer URL nicht zulässig sind, prozentkodiert. Außerdemwerden das Zeichen für den Leerschritt
und das Prozentzeichen (%) kodiert.

Darüber hinaus werden nicht reservierte Zeichen gemäß RCF3986 und Zeichen oberhalb von US-
ASCII x'7F', zum Beispiel deutsche Umlaute, von dieser Funktion kodiert.

SAGENCE benötigt Natural-Unicode-Unterstützung.

BeschreibungObjekt

Dies ist der Aufruf der erweiterten Kodierfunktion.SAGENCE

Parameter:

P-DEC-STR-E (U)
P-RET (I4) OPTIONAL /* 0: ok

/* else: Natural error returned
/* by the GIVING clause of
/* MOVE ENCODED.
/* This is the error which
/* comes up when a character
/* cannot be converted into
/* the target code page.

/* Error strategy:
/* Step 1: If a character shall be %-encoded and is not available
/* in the code page for %-encoding, the character will not be
/* %-encoded. It will be copied.
/* Step 2: If a character will not be %-encoded but copied from the
/* input format U-variable to a format A-variable (in *CODEPAGE)
/* and the character is not available in *CODEPAGE, a substitution
/* character will be used instead. The substitution character will
/* be %-encoded.
/* The last error will be returned in P-RET.
P-ERR-CHAR (U1) OPTIONAL /* Character causing the error
P-SPACE (A1) OPTIONAL /* '%' => %20

/* ' ' => ' '
/* else => '+' (default)

P-UNRES (A1) OPTIONAL /* 'E' => encode

BeschreibungObjekt

/* else => don't encode (default)
P-RES (A1) OPTIONAL /* 'E' => encode (default)

/* else => don't encode
P-OTHER (A1) OPTIONAL /* 'E' => encode (default)

/* else => don't encode
P-CP (A64) OPTIONAL /* IANA name e.g. UTF-8 (default)

/* or ISO-8859-1
/* On mainframe only code page names defined with the macro NTCPAGE
/* in the source module NATCONFG can be used. Other code page names
/* are rejected with a corresponding runtime error.
/*
P-CP-TABLE-CHAR(U1/1:*) OPTIONAL /* user selected char to be

/* %-encoded, e.g. 'ö' or '/'
P-CP-TABLE-ENC (A12/1:*) OPTIONAL /* user %-encoding

/* e.g. character 'ö'
/* '%F6' -> ISO-8859-1
/* '%C3%B6' -> UTF-8
/* e.g. character '/'
/* '/' -> '/' not encoded
/* although P-RES = 'E'

/* Characters in this table will be encoded according to the
/* specifed %-encoding. If the U12 encoding part is blank (space
/* according to *CODEPAGE) or the P-CP-TABLE-ENC value is equal to
/* the character, then the character will not be encoded at all.
/*

Der Copycode, der die Prototyp-Definition enthält, wird nur bei der Kompilierung benutzt,
um den Typ der Rückgabevariablen als Referenz für den Funktionsaufruf zu bestimmen und
um die Parameter zu prüfen, falls dies gewünscht wird.

SAGENCEP ist optional.

SAGENCEP

Beispiel-Programm in der Library SYSEXPG.URLX01

Beispielaufrufe

Verwendet werden die Standardvorgabewerte:

#URL-ENC := SAGENCE(<#URL-DEC-U>)

Alle Parameter, die möglich sind, werden angegeben:

#URL-ENC := SAGENCE(<#URL-DEC-U,L-RET,L-ERR-CHAR,L-SPACE,L-UNRES, ↩
L-RES,L-OTHER,L-CP,L-CP-TAB-CHAR(*),L-CP-TAB-ENC(*) >)

SAGDECE - Erweiterte Dekodierung (Format A nach Format U, optionale Parameter)

Die Funktion SAGDECE dient zum Dekodieren der von der Funktion SAGENCE gelieferten Prozent-
kodierungen. Falls ein Leerschrittzeichen und/oder eine Codepage angegeben werden, müssen
die Werte mit den bei der Kodierung angegebenen Werte übereinstimmen.

SAGDECE benötigt Natural-Unicode-Unterstützung.

BeschreibungObjekt

Dies ist der Aufruf der erweiterten Dekodierfunktion.SAGDECE

Parameter:

1 P-ENC-STR-E (A)
1 P-RET (I4) OPTIONAL /* 0: ok

/* else: Natural error returned
/* by the GIVING clause of
/* MOVE ENCODED.
/* This error comes up
/* when a %-encoded
/* character cannot be
/* converted into the
/* target code page.

/* The last error will be returned in P-RET.
1 P-ERR-CHAR (A12) OPTIONAL /* Error character %-encoded
1 P-SPACE (A1) OPTIONAL /* ' ' => ' '

/* else => '+' (default)
1 P-CP (A64) OPTIONAL /* IANA name e.g. UTF-8 (default)

/* or ISO-8859-1
/* On mainframe only code page names defined with the macro NTCPAGE
/* in the source module NATCONFG can be used. Other code page names
/* are rejected with a corresponding runtime error.
/*

Der Copycode, der die Prototyp-Definition enthält, wird nur bei der Kompilierung benutzt,
um den Typ der Rückgabevariablen als Referenz für den Funktionsaufruf zu bestimmen und
um die Parameter zu prüfen, falls dies gewünscht wird.

SAGDECEP ist optional.

SAGDECEP

Example program in der Library SYSEXPG.URLX01

Beispielaufrufe

Verwendet werden die Standardvorgabewerte:

BeschreibungObjekt

#URL-DEC-U := SAGDECE(<#URL-ENC>)

Alle Parameter, die möglich sind, werden angegeben:

#URL-DEC-U := SAGDECE(<#URL-ENC,L-RET,L-ERR-CHAR-DEC,L-SPACE,L-CP>)

Beispiel-Programm

Dieses Beispiel-Programm befindet sich in der Library SYSEXPG:

** Example 'URLX01': ENCODED-STR := SAGENC(<DECODED-STR>)
************************************************************************
DEFINE DATA
LOCAL
1 SAMPLE-STRING (A72)
/*
1 #URL-DEC (A) DYNAMIC
1 #URL-ENC (A) DYNAMIC
/*
1 #URL-DEC-U (U) DYNAMIC
/*
1 L-RET (I4) /* Return code
1 L-ERR-CHAR (U1) /* Error character
1 L-ERR-CHAR-DEC(A12) /* Decoded error character
1 L-SPACE (A1) /* '%' => %20, ' ' => ' ',

/* else => '+' (default)
1 L-UNRES (A1) /* 'E' => encode, else => don't encode (default)
1 L-RES (A1) /* 'E' => encode (default), else => don't encode
1 L-OTHER (A1) /* 'E' => encode (default), else => don't encode
1 L-CP (A64) /* default *CODEPAGE
1 L-CP-TAB-CHAR (U1/1:1)
1 L-CP-TAB-ENC (A12/1:1)
1 L-MSG (U72)
END-DEFINE
/*
/*
/*
WRITE 'Sample string to be processed:'
/* The string below shall be encoded and decoded again.
/* After decoding it should be unchanged.
SAMPLE-STRING := '"Decoded data!"'
WRITE SAMPLE-STRING (AL=72) /
/*
/* Assign the sample string to the input variable #URL-DEC of the
/* simple encoding function.
#URL-DEC := SAMPLE-STRING
/*
/* Copycode SAGENCP containing the prototype definition is used at

/* compilation time only in order to determine the type of the return
/* variable for function call reference and to check the parameters,
/* if this is desired. SAGENCP is optional.
INCLUDE SAGENCP
/*
/* SAGENC(<#URL-DEC>) is the simple encoding function call.
/*
/* Function SAGENC %-encodes a string to code page ISO-8859-1.
/* According to standard RFC3986 reserved characters and characters
/* below US-ASCII x'7F' which are not allowed in a URL will be
/* %-encoded.
/* Also the space and the percent sign will be encoded.
/* Unreserved characters according to RCF3986 and characters above
/* US-ASCII x'7F' will not be encoded. If you want to encode such
/* characters, use the extended encoding function.
/*
/* ---- Space ' ' -> '+'
/* ---- Percent sign '%' -> '%25'
/*
/* Unreserved characters according to RFC3986 (will not be encoded!):
/* ---- Period (fullstop) '.' -- '%2E'
/* ---- Tilde '~' -- '%7E'
/* ---- Hyphen '-' -- '%2D'
/* ---- Underscore character '_' -- '%5F'
/* ---- digits, lower and upper case characters
/* ---- 0123456789abcdefghijklmnopqrstuvwxyzABCDEFGHIJKLMNOPQRSTUVWXYZ
/*
/* Reserved characters according to RFC3986:
/* ---- Exclamation mark '!' -> '%21'
/* ---- Number sign '#' -> '%23'
/* ---- Dollar sign '$' -> '%24'
/* ---- Ampersand '&' -> '%26'
/* ---- Apostrophe ''' -> '%27'
/* ---- Left parenthesis '(' -> '%28'
/* ---- Right parenthesis ')' -> '%29'
/* ---- Asterisk '*' -> '%2A'
/* ---- Plus sign '+' -> '%2B'
/* ---- Comma ',' -> '%2C'
/* ---- Reverse solidus (backslash) '/' -> '%2F'
/* ---- Colon ':' -> '%3A'
/* ---- Semi-colon ';' -> '%3B'
/* ---- Equals sign '=' -> '%3D'
/* ---- Question mark '?' -> '%3F'
/* ---- Commercial at '@' -> '%40'
/* ---- Square bracket open '[' -> '%5B'
/* ---- Square bracket close ']' -> '%5D'
/*
/* Other characters below x'7F' (US-ASCII) but not allowed in URL
/* ---- Quotation mark '"' -> '%22'
/* ---- Less than '<' -> '%3C'
/* ---- Greater than '>' -> '%3E'
/* ---- Reverse solidus (backslash) '\' -> '%5C'

/* ---- Accent, Circumflex '^' -> '%5E'
/* ---- Accent, Grave '`' -> '%60'
/* ---- Opening brace '{' -> '%7B'
/* ---- Vertical bar '|' -> '%7C'
/* ---- Closing brace '}' -> '%7D'
/*
#URL-ENC := SAGENC(<#URL-DEC>)
/*
/*
WRITE 'Simple function, encoded:'
WRITE #URL-ENC (AL=72)
/*
/* Copycode SAGDECP containing the prototype definition is used at
/* compilation time only in order to determine the type of the return
/* variable for function call reference and to check the parameters,
/* if this is desired. SAGDECP is optional.
INCLUDE SAGDECP
/*
/* SAGDEC(<#URL-ENC>) is the simple decoding function call.
/* It decodes the above described %-encodings.
/*
#URL-DEC := SAGDEC(<#URL-ENC>)
/*
/*
/* The result after encoding and decoding must be equal to the original
/* SAMPLE-STRING.
WRITE 'Simple function, decoded:'
WRITE #URL-DEC (AL=72)
/*
/*
/*
WRITE /
/*
/*
/*
/* Assign the sample string to the input variable #URL-DEC-U of the
/* enhanced encoding function.
#URL-DEC-U := SAMPLE-STRING
/*
/* Copycode SAGENCEP containing the prototype definition is used at
/* compilation time only in order to determine the type of the return
/* variable for function call reference and to check the parameters,
/* if this is desired. SAGENCEP is optional.
INCLUDE SAGENCEP
/*
/* This is the enhanced encoding function call.
/* The way, characters will be %-encoded dependes on the input
/* parameter of the function.
/* The parameters of the encoding and decoding function are preset
/* with the default values.
/* L-CP-TAB-CHAR(*) and L-CP-TAB-ENC(*) don't have default values.
/* L-CP-TAB-CHAR(1) = 'ä' and L-CP-TAB-ENC(1) = '%C3%A4' will not be

65Systemfunktionen

Funktionen, die als Natural-Objekte ausgeliefert werden



/* used for the sample string '"Decoded data!"'. The string does not
/* contain an 'ä'.
L-SPACE := '+' /* encoding and decoding
L-UNRES := 'D' /* encoding only
L-RES := 'E' /* encoding only
L-OTHER := 'E' /* encoding only
L-CP := 'UTF-8' /* encoding and decoding

/* e.g. ISO-8859-1, UTF-16BE, UTF-32BE
L-CP-TAB-CHAR(1) := 'ä' /* encoding only
L-CP-TAB-ENC (1) := '%C3%A4' /* encoding only
/*
/* Note that all possible parameters are specified for this sample
/* call.
/* If the default values shall be used and no return code is wanted,
/* all parameters can be omitted, besides the string #URL-DEC-U.
/*
#URL-ENC := SAGENCE(<#URL-DEC-U,L-RET,L-ERR-CHAR,L-SPACE,L-UNRES,

L-RES,L-OTHER,L-CP,L-CP-TAB-CHAR(*),L-CP-TAB-ENC(*) >)
WRITE 'Extended function, encoded:'
WRITE #URL-ENC (AL=72)
IF L-RET NE 0 THEN

/* If L-RET = 0, the function worked ok. Else L-RET contains the
/* Natural error returned by the GIVING clause of MOVE ENCODED.
/* The error comes up when a character cannot be converted into
/* the target codepage, e.g. because a character does not exist
/* in the target codepage.
COMPRESS 'Error' L-RET 'with MOVE ENCODED of' L-ERR-CHAR INTO L-MSG
WRITE L-MSG

END-IF
/*
/* Copycode SAGDECEP containing the prototype definition is used at
/* compilation time only in order to determine the type of the return
/* variable for function call reference and to check the parameters,
/* if this is desired. SAGDECEP is optional.
INCLUDE SAGDECEP
/*
/* This is the 1st enhanced decoding function call with 5 parameters.
/* Note that all possible parameters are specified for this sample
/* call.
/* Since the parameters have the default values, the subsequent
/* function calls return the same result although parameters
/* have been omitted.
#URL-DEC-U := SAGDECE(<#URL-ENC,L-RET,L-ERR-CHAR-DEC,L-SPACE,L-CP>)
WRITE 'Extended function, decoded:'
WRITE #URL-DEC-U (AL=72)
IF L-RET NE 0 THEN

/* If L-RET = 0, the function worked ok. Else L-RET contains the
/* Natural error returned by the GIVING clause of MOVE ENCODED.
/* The error comes up when a %-encoded character cannot be converted
/* into the target codepage, e.g. because a character does not exist
/* in the target codepage.
COMPRESS 'Error' L-RET 'with MOVE ENCODED of' L-ERR-CHAR INTO L-MSG

WRITE L-MSG
RESET L-RET

END-IF
/*
/* This is the 2nd enhanced decoding function call with one parameter.
#URL-DEC-U := SAGDECE(<#URL-ENC>)
WRITE #URL-DEC-U (AL=72)
/* L-RET will not be returned
/*
/* This is the 3rd enhanced decoding function call with 3 parameters.
#URL-DEC-U := SAGDECE(<#URL-ENC,L-RET,2X,L-CP>)
WRITE #URL-DEC-U (AL=72)
IF L-RET NE 0 THEN

COMPRESS 'Error' L-RET 'with MOVE ENCODED of' L-ERR-CHAR INTO L-MSG
WRITE L-MSG
RESET L-RET

END-IF
/*
END

Base64-Kodierung

Dieser Abschnitt beschreibt Natural-Funktionen, die Sie benutzen können, um mittels Base64-
Konvertierung Binärdaten in druckbare, netzkompatible Daten bzw. in umgekehrter Richtung zu
konvertieren.

Base64-Konvertierung bedeutet eine Umwandlung vom Format B in das Format A und zurück
nach Format B. Dabei werden 6 (binäre) Bits in 8 (alphanumerische) Bits umgewandelt. Beispiel:
Ein Wert mit Format/Länge B3 wird in einen Wert mit Format/Länge A4 umgewandelt.

Anmerkung: Jeder Binärwertwird in einen eindeutigen alphanumerischenWert umgewandelt.
Die Rückumwandlung dieses alphanumerischen Werts ergibt dann wieder den ursprüng-
lichen Binärwert. Dies ist jedoch bei denmeistenUmwandlungen von FormatAnach Format
B und zurück nach Format A nicht der Fall.

Die Konvertierung kann benutzt werden, um eine Datei vom Typ .bmp per TCP/IP zu übertragen
oder um Natural-Binär- oder Ganzzahlwerte per Utility-Protokoll zu übertragen.

Nur bei Open Systems: Es stehen drei Modi zur Verfügung: RFC3548, RFC2045 und NATRPC
(Standardeinstellung). NATRPC bedeutet, dass die Umwandlung entsprechend der NATRPC-Ligik
erfolgt. Diese ist zu 100% großrechnerkompatibel. RFC2045 ist der Standardvorgabewert für den
CMBASE64-Aufruf. RFC3548 ist wie NATRPC, aber nicht benötigte alphanumerische Bytes werden
mit einem Gleichheitszeichen (=) gefüllt.

Folgende Funktionen stehen zur Verfügung:

■ SAG64BA - Konvertierung von binär nach alphanumerisch

67Systemfunktionen

Funktionen, die als Natural-Objekte ausgeliefert werden



■ SAG64AB - Konvertierung von alphanumerisch nach binär

Diese beiden Funktionen zusammengenommen bieten die gleiche Funktionalität wie das API
USR4210N. Dieses API befindet sich in der Library SYSEXT.

SAG64BA - Konvertierung von binär nach alphanumerisch

Die Funktion SAG64BAwandelt mittels Base64-Konvertierung Binärdaten in druckbare, netzkom-
patible Daten

BeschreibungObjekt

Dies ist die Funktion zur Wandlung vom Binärformat in alphanumerisches Format.SAG64BA

Parameter:

1 PARM-B (B) DYNAMIC BY VALUE
/* Binary source input/target output

1 PARM-RC (I4) OPTIONAL
/* 0: ok
/* Mainframe
/* 1 Source is not numeric
/* 2 Source is not packed
/* 3 Source is not floating point
/* 4 Overflow, source doesn't fit into target
/* 5 Integer overflow
/* 6 Source is not a valid date or time
/* 7 Length error (hex input not even)
/* 8 Target precision is less than source precision
/* 9 Float underflow (result->0)
/* 10 Alpha source contains non-hex characters
/* 20 Invalid function code
/* Open Systems
/* 1 Invalid value for RFC parameter
/* 2 Invalid function code
/* 3 CMBASE64: Overflow, source doesn't fit into
/* target
/* 4 CMBASE64: Non-base64 character found in encoded
/* data
/* 5 CMBASE64: Out of memory
/* 6 CMBASE64: Invalid number of parameters
/* 7 CMBASE64: Invalid parameter type
/* 8 CMBASE64: Invalid parameter length
/* 9 CMBASE64: Invalid function code
/* 10 CMBASE64: Unkown return code

1 PARM-ERRTXT (A72) OPTIONAL
/* blank, if ok no error
/* else error text

1 PARM-RFC (B1) OPTIONAL
/* OS only, not used for MF
/* 0 - RFC3548; 3 - RFC2045; 4 - NATRPC;

BeschreibungObjekt

Der Copycode, der die Prototyp-Definition enthält, wird nur bei der Kompilierung benutzt,
um den Typ der Rückgabevariablen als Referenz für den Funktionsaufruf zu bestimmen und
um die Parameter zu prüfen, falls dies gewünscht wird.

SAG64BAP ist optional.

SAG64BAP

Beispiel-Programm in der Library SYSEXPG.

Verwendet werden die Standardvorgabewerte:

B64X01

PARM-A := SAG64BA(<PARM-B>)

Alle Parameter, die möglich sind, werden angegeben (PARM-RFC gilt nicht bei Großrechnern):

PARM-A := SAG64BA(<PARM-B,PARM-RC,PARM-ERRTXT,PARM-RFC>)

SAG64AB - Wandlung vom alphanumerischen Format in das Binärformat

Die Funktion SAG64ABwandelt mittels Base64-Konvertierung druckbare, netzkompatible Daten
in Binärdaten.

BeschreibungObjekt

Dies ist die Funktion zur Wandlung vom alphanumerischen Format in das Binärformat.SAG64AB

Parameter:

1 PARM-A (A)
/* Alpha source input/target output

1 PARM-RC (I4) OPTIONAL
/* 0: ok
/* Mainframe
/* 1 Source is not numeric
/* 2 Source is not packed
/* 3 Source is not floating point
/* 4 Overflow, source doesn't fit into target
/* 5 Integer overflow
/* 6 Source is not a valid date or time
/* 7 Length error (hex input not even)
/* 8 Target precision is less than source precision
/* 9 Float underflow (result->0)
/* 10 Alpha source contains non-hex characters
/* 20 Invalid function code
/* Open Systems
/* 1 Invalid value for RFC parameter
/* 2 Invalid function code
/* 3 CMBASE64: Overflow, source doesn't fit into
/* target
/* 4 CMBASE64: Non-base64 character found in encoded
/* data

BeschreibungObjekt

/* 5 CMBASE64: Out of memory
/* 6 CMBASE64: Invalid number of parameters
/* 7 CMBASE64: Invalid parameter type
/* 8 CMBASE64: Invalid parameter length
/* 9 CMBASE64: Invalid function code
/* 10 CMBASE64: Unkown return code

1 PARM-ERRTXT (A72) OPTIONAL
/* blank, if ok no error
/* else error text

1 PARM-RFC (B1) OPTIONAL
/* OS only, not used for MF
/* 0 - RFC3548; 3 - RFC2045; 4 - NATRPC;

Der Copycode, der die Prototyp-Definition enthält, wird nur bei der Kompilierung benutzt,
um den Typ der Rückgabevariablen als Referenz für den Funktionsaufruf zu bestimmen und
um die Parameter zu prüfen, falls dies gewünscht wird.

SAG64ABP ist optional.

SAG64ABP

Beispiel-Programm in der Library SYSEXPG.

Verwendet werden die Standardvorgabewerte:

B64X01

PARM-B := SAG64AB(<PARM-A>)

Alle Parameter, die möglich sind, werden angegeben (PARM-RFC gilt nicht bei Großrechnern):

PARM-B := SAG64AB(<PARM-A,PARM-RC,PARM-ERRTXT,PARM-RFC>)

Beispiel-Programm

Beispiel-Programm B64X01 in der Library SYSEXPG:

** Example 'B64X01': BASE64-A-STR := SAG64BA(<BASE64-B-STR>)
************************************************************************
* Function ......... Convert binary data into printable,
* network-compatible data or vice versa using
* Base64 encoding.
*
* Base64 encoding means (B) -> (A) -> (B),
* where 6 (binary) bits will be encoded into 8
* (alpha) bits, e.g a (B3) value will be encoded
* into a (A4) value.
*
* Note: Every binary value will be encoded into
* a non-ambiguous alpha value. Re-encoding this
* alpha value again will result in the original
* binary value. However, this is not the case with
* most of the (A) -> (B) -> (A) encodings.

*
* The encoding may be used to transfer a .bmp
* file via TCP/IP, or to transfer Natural binary or
* integer values via the utility protocol.
*
* Open Systems only:
* On Open Systems, there are 3 modes:
* RFC3548, RFC2045 and NATRPC (default).
* NATRPC means the encoding follows
* the NATRPC logic. This is 100% MF compatible.
* RFC2045 is the default of the CMBASE64 call.
* RFC3548 is like NATRPC, but alpha bytes not
* needed are filled with '='.
*
DEFINE DATA
LOCAL
1 FUNCTION (A2)

/* 'AB' Alpha to binary encoding
/* 'BA' Binary to alpha encoding

1 PARM-RC (I4)
/* 0: ok
/* Mainframe
/* 1 Source is not numeric
/* 2 Source is not packed
/* 3 Source is not floating point
/* 4 Overflow, source doesn't fit into target
/* 5 Integer overflow
/* 6 Source is not a valid date or time
/* 7 Length error (hex input not even)
/* 8 Target precision is less than source precision
/* 9 Float underflow (result->0)
/* 10 Alpha source contains non-hex characters
/* 20 Invalid function code
/* Open Systems
/* 1 Invalid value for RFC parameter
/* 2 Invalid function code
/* 3 CMBASE64: Overflow, source doesn't fit into
/* target
/* 4 CMBASE64: Non-base64 character found in encoded
/* data
/* 5 CMBASE64: Out of memory
/* 6 CMBASE64: Inalid number of parameters
/* 7 CMBASE64: Invalid parameter type
/* 8 CMBASE64: Invalid parameter length
/* 9 CMBASE64: Invalid function code
/* 10 CMBASE64: Unkown return code

1 PARM-ERRTXT (A72)
/* blank, if ok no error
/* else error text

1 PARM-A (A) DYNAMIC
/* Alpha source input/target output

1 PARM-B (B) DYNAMIC

71Systemfunktionen

Funktionen, die als Natural-Objekte ausgeliefert werden



* /* Binary source input/target output
1 PARM-RFC (B1)

/* OS only, not used for MF
/* 0 - RFC3548; 3 - RFC2045; 4 - NATRPC;

/*
1 #BACKUP-A (A) DYNAMIC
1 #BACKUP-B (B) DYNAMIC
END-DEFINE
/*
/*
SET KEY ALL
/*
/* Copycode SAG64BAP and SAG64ABP containing the prototype definition
/* is used at compilation time only in order to determine the type of
/* the return variable for function call reference and to check the
/* parameters, if this is desired. SAG64BAP and SAG64ABP are optional.
INCLUDE SAG64BAP
INCLUDE SAG64ABP
/*
REPEAT

RESET PARM-A PARM-B
REDUCE DYNAMIC PARM-A TO 0
REDUCE DYNAMIC PARM-B TO 0
FUNCTION := 'BA'
PARM-B := H'0123456789ABCDEF'
INPUT (AD=MIL IP=OFF CD=NE) WITH TEXT PARM-ERRTXT
// 10T 'Base64 Encoding:' (YEI)
/ 10T '-' (19) (YEI) /
/ 10T 'Function (BA,AB) ..' (TU) FUNCTION (AD=T)
/ 10T 'Alpha In/Output ...' (TU) PARM-A (AL=30)
/ 10T 'Binary In/Output ..' (TU) PARM-B (EM=HHHHHHHH)
/ 10T 'Response ..........' (TU) PARM-RC (AD=OD CD=TU)
/ PARM-ERRTXT (AD=OD CD=TU)

RESET PARM-ERRTXT
IF *PF-KEY NE 'ENTR'

ESCAPE BOTTOM
END-IF
/*
RESET #BACKUP-A #BACKUP-B
REDUCE DYNAMIC #BACKUP-A TO 0
REDUCE DYNAMIC #BACKUP-B TO 0
#BACKUP-A := PARM-A
#BACKUP-B := PARM-B
/*
IF FUNCTION = 'BA'

/* Parameter PARM-RC, PARM-ERRTXT and PARM-RFC are optional
/* Parameter PARM-RFC does not apply to mainframe
/* PARM-A := SAG64BA(<PARM-B,PARM-RC,PARM-ERRTXT,PARM-RFC>)
PARM-A := SAG64BA(<PARM-B,PARM-RC,PARM-ERRTXT>)
/* PARM-A := SAG64BA(<PARM-B,PARM-RC>)
/* PARM-A := SAG64BA(<PARM-B>)

ELSE

/* Parameter PARM-RC, PARM-ERRTXT and PARM-RFC are optional
/* Parameter PARM-RFC does not apply to mainframe
/* PARM-B := SAG64AB(<PARM-A,PARM-RC,PARM-ERRTXT,PARM-RFC>)
PARM-B := SAG64AB(<PARM-A,PARM-RC,PARM-ERRTXT>)
/* PARM-B := SAG64AB(<PARM-A,PARM-RC>)
/* PARM-B := SAG64AB(<PARM-A>)

END-IF
/*
IF PARM-RC NE 0 THEN

WRITE 'Encoding' FUNCTION
WRITE NOTITLE PARM-ERRTXT

ELSE
IF FUNCTION = 'BA' THEN
WRITE 'Binary -> Alpha'
WRITE '=' PARM-B (EM=HHHHHHHHHHHHHHHHHHHHHHHHH)

/ '=' PARM-A (AL=50)
RESET PARM-B
REDUCE DYNAMIC PARM-B TO 0
FUNCTION := 'AB'

ELSE
WRITE 'Alpha -> Binary'
WRITE '=' PARM-A (AL=50) /

'=' PARM-B (EM=HHHHHHHHHHHHHHHHHHHHHHHHH)
RESET PARM-A
REDUCE DYNAMIC PARM-A TO 0
FUNCTION := 'BA'

END-IF
/*
IF FUNCTION = 'BA'
/* Parameter PARM-RC, PARM-ERRTXT and PARM-RFC are optional
/* Parameter PARM-RFC does not apply to mainframe
/* PARM-A := SAG64BA(<PARM-B,PARM-RC,PARM-ERRTXT,PARM-RFC>)
PARM-A := SAG64BA(<PARM-B,PARM-RC,PARM-ERRTXT>)
/* PARM-A := SAG64BA(<PARM-B,PARM-RC>)
/* PARM-A := SAG64BA(<PARM-B>)

ELSE
/* Parameter PARM-RC, PARM-ERRTXT and PARM-RFC are optional
/* Parameter PARM-RFC does not apply to mainframe
/* PARM-B := SAG64AB(<PARM-A,PARM-RC,PARM-ERRTXT,PARM-RFC>)
PARM-B := SAG64AB(<PARM-A,PARM-RC,PARM-ERRTXT>)
/* PARM-B := SAG64AB(<PARM-A,PARM-RC>)
/* PARM-B := SAG64AB(<PARM-A>)

END-IF
IF PARM-RC NE 0 THEN
WRITE 'Encoding' FUNCTION
WRITE NOTITLE PARM-ERRTXT

ELSE
IF FUNCTION = 'BA' THEN

WRITE 'Binary -> Alpha'
WRITE '=' PARM-B (EM=HHHHHHHHHHHHHHHHHHHHHHHHH)
/ '=' PARM-A (AL=50)

IF PARM-A = #BACKUP-A THEN

WRITE '******** Encoding successful ********'
ELSE

WRITE '******** Value changed by encoding ********'
END-IF

ELSE
WRITE 'Alpha -> Binary'
WRITE '=' PARM-A (AL=50) /

'=' PARM-B (EM=HHHHHHHHHHHHHHHHHHHHHHHHH)
IF PARM-B = #BACKUP-B THEN

WRITE '******** Encoding successful ********'
ELSE

WRITE '******** Value changed by encoding ********'
END-IF

END-IF
END-IF

END-IF
END-REPEAT
END

